
Modifier des équipements

Lorsqu’on crée un équipement, on n’est pas toujours sûr qu’il ne faudra pas le retoucher par 
la suite jusqu’à la fin du projet. De ce fait, il faut pouvoir le modifier tout au long du projet.

Si l’équipement a été créé depuis l’éditeur d’équipement, il sera possible de modifier sa forme 
et ses ajustements juste en sélectionnant l’équipement et par le clic droit de la souris de 
sélectionner "Modifier équipement" ou en tapant directement la commande "PLANTEQUIP-
MENTMODIFY.

Si l’équipement provient d’une convertion d’objets AutoCAD 3D (de préférence) il sera possible 
de le modifier par la suite par l’éditeur de bocs à condition que l’objet d’origine était un bloc.

Après la modification du bloc, il faudra redéfinir correctement les piquages modifiés.

Par contre si les objets d’origines n’avaient pas été regroupés dans un bloc la modification 
n’aurait été possible qu’en décomposant l’équipement. De ce fait après sa modification il 
aurait fallu reconvertir les objets en équipement puis recréer tous les piquages un à un.
Même principe pour un équipement provenant d’un objet Inventor.

Le problème majeur vient dans le cas où il est nécessaire de décomposer l’équipement pour 
le modifier, autant dans sa forme que dans la position des piquages.

Ce problème vient du fait de la conversion d’un objet en équipement. AutoCAD Plant 3D 
mémorise cet équipement sous un nom de bloc dit "anonyme". Un bloc "anonyme" n’est pas 
visible dans la liste des blocs donc impossible de le modifier depuis l’éditeur de blocs.
Un bloc "anonyme" ne peut pas être renommé, normal puisque son nom n’est pas visible !

La seule solution est de passer par de la programmation !

Pas de crainte, votre serviteur vous a créé un petit programme pour traiter ce cas. 
Programme à utiliser avec modération !

Voici donc le code du programme qui permet de renommer un bloc dit "anonyme" en un bloc 
visible et ainsi pouvoir le modifier dans l’éditeur de blocs.

(defun c:AP3D_ConvertBlocEquip (/ Selection NomDuBloc NouveauNomDuBloc)

  (Setq Selection (entget (car (entsel))))

  (if (= (cdr (assoc 0 Selection)) "ACPPEQUIPMENT")
    (progn

      (Setq
	 NomDuBloc (cdr (assoc 2 (entget (cdr (assoc 340 Selection)))))
      )

  (if (= (substr NomDuBloc 1 1) "*")
	 (progn
	   (setq	 NouveauNomDuBloc (getstring (strcat "\nNom actuel <"NomDuBloc"> / Nouveau nom : ") )
	   (RenommeBlocAnonyme NomduBloc NouveauNomDuBloc)
	 )
	 (command "_-Bedit" NomduBloc)
	 )
    )
    (princ "\nCet objet n’est pas un équipement.")
  )

  (princ)
)

(defun RenommeBlocAnonyme (NomduBloc NouveauNomDuBloc 
					      / acad_app  acad_doc acad_blocs Bloc Choix)
  (vl-load-com)
  (setq	 acad_app   (vlax-get-acad-object)
	            acad_doc   (vla-get-ActiveDocument acad_app)
	            acad_blocs (vla-get-blocks acad_doc)
  )

  (setq bloc (vla-item acad_blocs NomduBloc))

  (while (tblsearch "BLOCK" NouveauNomDuBloc)
    (setq NouveauNomDuBloc (strcat "$" NouveauNomDuBloc))
  )

  (vla-put-name bloc NouveauNomDuBloc)

  (princ
    (strcat "\nBloc " NomduBloc " renommé en " NouveauNomDuBloc)
  )

  (initget "O N" 7)
  (setq choix (getkword "\nOuvrir l’éditeur de bloc ? [O N]:"))

  (if (= choix "O")
    (command "_-Bedit" NouveauNomDuBloc)
  )

  (princ)

)

Fonctionnement :

Après avoir chargé le programme tapez la commande "AP3D_ConvertBlocEquip". 

Suivez maintenant les étapes :

Etape 1

Sélection de l’équipement
Saisisez un nom de bloc
Tapez "O" pour accéder à l’éditeur de blocs.

Etape 2

Dans l’éditeur de bloc modifiez l’équipement
Après modification sortez de l’éditeur de 
blocs en sauvegardant les modifications.
Une fois revenu dans le dessin, l’équipement 
est mis à jour. 

Etape 3

Si les piquages ont été modifiés, n’oubliez pas 
de les redéfinir au niveau de l’équipement.

Pas plus compliqué que ça !

Ci-après l’adresse d’une vidéo montrant le 
fonctionnement de l’application

https://autode.sk/37xJiAM

